


Supplementary Materials for

**A New Radiocarbon Sequence from Lamanai, Belize:  
Two Bayesian Models from One of Mesoamerica's Most Enduring Sites**

Jonathan A. Hanna,\* Elizabeth Graham, David M. Pendergast, Julie A. Hoggarth,  
David L. Lentz, Douglass J. Kennett

\*Corresponding author, email: jah1147@psu.edu

# S1. Plot of All Radiocarbon Dates (from final models unless noted)



# S2: OxCal v.4.2.4 CQL Programming Codes

## Structures N10-77 and N10-12

```
Plot("N10-77_and_N10-12_with_Outliers")
{
  Outlier_Model("General",T(5),U(0,4),"t");
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"c");
  Outlier_Model("SSimple",N(0,2),0,"s");
  Sequence("Room C")
  {
    Boundary("Floor 2");
    Phase("Within Floor 2 (Sealed by Floor 1)")
    {
      R_Date("LA 2524 [OxA-18017(13)]", 1275, 26)
      {
        Outlier("General", 0.05);
        color="Orange";
      };
      R_Date("LA 2532 [OxA-18020(15)]", 1240, 26)
      {
        Outlier("General", 0.05);
        color="Orange";
      };
      R_Date("LA 2522 [OxA-18016(12)]", 1260, 26)
      {
        Outlier("General", 0.05);
        color="Orange";
      };
    };
    Boundary("Floor 2 End");
    Phase("Cut into Floor 1")
    {
      Combine("Floor 1, Cache 13")
      {
        Outlier("Charcoal", 0.05);
        color="Orange";
        R_Date("LA 2525 [OxA-18019(14)]", 1282, 26)
        {
          Outlier("SSimple", 0.05);
        };
        R_Date("LA 2525 [OxA-18018(14)]", 1331, 27)
        {
          Outlier("SSimple", 1);
        };
      };
      R_Date("LA 1798 [OxA-17976(9)]", 1284, 25)
      {
        Outlier("General", 0.05);
        color="Orange";
      };
    };
    Boundary("Floor 1 End");
    Sequence("Room B2")
    {
      Boundary("Penultimate Floor");
      Phase("Sealed by Final Floor, Start of late LC")
      {
        Combine("Cache 2")
        {
          Outlier("General", 0.05);
          color="Orange";
          R_Date("LA 1777 [OxA-17985(3)]", 1402, 25)
          {
            Outlier("SSimple", 0.05);
          };
          R_Date("LA 1777 [OxA-17970(3)]", 1409, 25)
          {
            Outlier("SSimple", 0.05);
          };
        };
        R_Date("LA 1784 [OxA-17974(7)]", 1304, 25)
        {
          Outlier("General", 0.05);
          color="Orange";
        };
        R_Date("LA 1785/1 [OxA-17975(8)]", 1297, 25)
        {
          Outlier("General", 0.05);
          color="Orange";
        };
        R_Date("LA 1783 [OxA-17973(6)]", 1280, 24)
        {
          Outlier("General", 0.05);
          color="Orange";
        };
      };
      Boundary("End Penultimate Floor-- Start of Terminal Classic");
      After()
      {
        XReference("Floor 1 End");
      };
      R_Date("LA 1764 [OxA-17969(2)]", 1312, 25)
      {
        Outlier("General", 0.05);
        color="Green";
      };
    };
    Boundary("N10-77 Termination");
  };
  Sequence("N10-12")
  {
    After()
    {
      XReference("N10-77 Termination");
    };
    Phase("N10-12 Cache 2")
    {
      R_Date("LA 1894/8 [OxA-18015(11)]", 1206, 26)
      {
        Outlier("General", 0.05);
        color="Green";
      };
      R_Date("LA 1894/6 [OxA-18014(10)]", 1282, 26)
      {
        Outlier("Charcoal", 1);
        color="Green";
      };
    };
    Boundary("Early PC");
    R_Date("LA 1742 [OxA-17968(1)]", 1050, 24)
    {
      Outlier("General", 0.05);
      color="Blue";
    };
    Boundary("End Occupation");
    C_Date("Spanish Contact", 1544, 0);
  };
};
```

## Structure N10-2

```
Plot("N10-2_with_Outliers")
{
  Outlier_Model("General",T(5),U(0,4),"t");
  Outlier_Model("SSimple",N(0,2),0,"s");
  Outlier_Model("Charcoal",Exp(1,-10,0),U(0,3),"c");
  Sequence("N10-2")
  {
    Boundary("Boundary 1: Begin Sequence");
    R_Date("LA 110/1C-Ork [GX-4662]", 1235, 130)
    {
      Outlier("General", 0.05);
    };
    Boundary("Boundary 2: Transition");
    R_Date("LA 115/2C-Tok [GX-4664]", 1251, 129)
    {
      Outlier("General", 0.05);
    };
    Boundary("Boundary 3: Transition");
    Phase("Gom Phase")
    {
      R_Date("LA 30/1C-Gom [GX-4659]", 1786, 139)
      {
        Outlier("Charcoal", 1);
      };
      R_Date("LA 177/1C-Gom [GX-4670]", 1061, 124)
      {
        Outlier("General", 0.05);
        color="Blue";
      };
    };
    Combine("LA 115/1C-Gom")
    {
      Outlier("General", 0.5);
      color="Blue";
      R_Date("LA 115/1C-Gom [GX-4663]", 715, 130)
      {
        Outlier("SSimple", 0.5);
      };
      R_Date("LA 115/1C-Gom [OxA-18022 (18)]", 950, 25)
      {
        Outlier("SSimple", 1);
      };
    };
    Combine("N10-2-4th, Cache 2")
    {
      Outlier("General", 0.05);
      color="Blue";
      R_Date("LA 34/1C-Gom [GX-4660]", 1040, 115)
      {
        Outlier("SSimple", 1);
      };
      R_Date("LA 34/1C-Gom [OxA-18021 (16)]", 856, 25)
      {
        Outlier("SSimple", 0.05);
      };
    };
    R_Date("LA 34/2C-Gom [GX-4661]", 830, 120)
    {
      Outlier("General", 0.05);
      color="Blue";
    };
    Boundary("Boundary 4: Transition");
    R_Date("LA 171/1C-Bat [GX-4669]", 1191, 129)
    {
      Outlier("Charcoal", 1);
    };
    Boundary("Boundary 5: Transition");
    R_Date("LA 139/1C-Fern [GX-4666]", 826, 134)
    {
      Outlier("General", 0.05);
    };
    Boundary("Boundary 6: End Sequence");
    C_Date("Spanish Contact", 1544, 0);
  };
};
```

## Trapezoidal Tzunun Phase

```
Plot("Lam_Tzunun_Trap")
{
  Sequence()
  {
    Boundary("MidStart")
    {
      Start("Start Start");
      Transition("Duration Start");
      End("End Start");
    };
    Phase()
    {
      Prior("LA 2524", "LA_2524_Tzunun_2.prior");
      Prior("LA 2532", "LA_2532_Tzunun_2.prior");
      Prior("LA 2522", "LA_2522_Tzunun_2.prior");
      Prior("LA 2525", "LA_2525_combined_Tzunun_2.prior");
      Prior("LA 1798", "LA_1798_Tzunun_2.prior");
      Prior("Cache 2 (LA 1777)", "Cache_2_LA1777_Tzunun_2.prior");
      Prior("LA 1784", "LA_1784_Tzunun_2.prior");
      Prior("LA 1785/1", "LA_1785_Tzunun_2.prior");
      Prior("LA 1783", "LA_1783_Tzunun_2.prior");
    };
    Boundary("Mid End")
    {
      Start("Start End");
      Transition("Duration End");
      End("End End");
    };
  };
};
```

## Trapezoidal Terclerp Phase

```
Plot("Lam_Terclerp_Trap")
{
  Sequence()
  {
    Boundary("MidStart")
    {
      Start("Start Start");
      Transition("Duration Start");
      End("End Start");
    };
    Phase()
    {
      Prior("LA 1764", "LA_1764_modelled_2.prior");
      Prior("LA 1894/8", "LA_1894_8_modelled_2.prior");
      Prior("LA 1894/6", "LA_1894_6_modelled_2.prior");
    };
    Boundary("Mid End")
    {
      Start("Start End");
      Transition("Duration End");
      End("End End");
    };
  };
};
```

## Trapezoidal Buk Phase

```
Plot("Lam_Buk_Trap")
{
  Sequence()
  {
    Boundary("MidStart")
    {
      Start("Start Start");
      Transition("Duration Start");
      End("End Start");
    };
    Phase()
    {
      Prior("LA 1742", "LA_1742_Buk_2.prior");
      Prior("LA 177/1C", "LA177_1C_Buk_2.prior");
      Prior("LA 115/1C Combined", "LA115_Combined_Buk_2.prior");
      Prior("LA 34/2C", "LA34_2C_Buk_2.prior");
      Prior("LA 34/1C (N10-2/2 Comb.)", "LA34_1C_Combined_Buk_2.prior");
    };
    Boundary("Mid End")
    {
      Start("Start End");
      Transition("Duration End");
      End("End End");
    };
  };
};
```

### Correction for Fractionation Effects

Only five samples were corrected by Geochron (GX-4660, GX-4661, GX-4662, GX-4663, and GX-4665), though the  $\delta^{13}\text{C}$  values for these were not provided in 1977. A worksheet provided on the Calib website (Stuiver and Reimer 2015) was used to reverse-calculate the  $\delta^{13}\text{C}$  values and arrive at the uncorrected ages. Conversely, corrections for the remaining Geochron samples were made using estimated  $\delta^{13}\text{C}$  values from Stuiver and Polach (1977:Fig. 1). See main text (“Evaluation of the 1976-77 Geochron Dates”) for further discussion.

### S3: Corrections for Lamanai’s Conventional Radiometric Samples

Sample No.	Lot No.	Material type	Uncorrected Radiocarbon Age, in $^{14}\text{C}$ years BP	Estimated $\delta^{13}\text{C}$ for samples measured as $^{14}\text{C}/^{12}\text{C}$ ratios	Uncertainty in Estimated $\delta^{13}\text{C}$ values	$\delta^{13}\text{C}$ Corrected Age, in $^{14}\text{C}$ years BP
GX-4659	N10-2, LA 30/1C	Wood charcoal fragments	1770 ± 135	-24†	± 2‰	1786 ± 139*
GX-4660	N10-2, LA 34/1C	Charred maize kernels and stalks ( <i>Zea mays</i> )	915 ± 110*	-17.4*	± 2‰	1040 ± 115
OXA-18021	N10-2, LA 34/1C	Charred maize kernels and cobs ( <i>Zea mays</i> ), likely harvested shortly before event; also <i>Pinus caribaea</i> and <i>Phaseolus vulgaris</i>	(ORAU re-run)	-9.6	± 0.2‰	856 ± 25
GX-4661	N10-2, LA 34/2C	Charred beans ( <i>Phaseolus vulgaris</i> ) and wood charcoal ( <i>Pinus caribaea</i> , <i>Areaceae</i> sp., angiosperm)—failed re-run at ORAU	840 ± 116*	-25.6*	± 2‰	830 ± 120
GX-4662	N10-2, LA 110/1C	‘Fibrous’ wood charcoal	1240 ± 126*	-25.3*	± 2‰	1235 ± 130
GX-4663	N10-2, LA 115/1C	Charcoal fragments; probably wattle and possibly some other wood as well	760 ± 126*	-27.7*	± 2‰	715 ± 130
OXA-18022	N10-2, LA 115/1C	Wood charcoal ( <i>Pinus caribaea</i> , <i>Acrocomia aculeata</i> )—young wood	(ORAU re-run)	-26.2	± 0.2‰	950 ± 25
GX-4664	N10-2, LA 115/2C	Coarse wood charcoal	1235 ± 125	-24†	± 2‰	1251 ± 129*
GX-4665	N10-2, LA 136/1C	Charcoal fragments, possibly palm wood	1710 ± 121*	-26.2*	± 2‰	1690 ± 125
GX-4666	N10-2, LA 139/1C	Coarse wood charcoal	810 ± 130	-24†	± 2‰	826 ± 134*
GX-4667	N10-7, LA 166	Wood charcoal	1510 ± 130	-24†	± 2‰	1526 ± 134*
GX-4668	N10-2, LA 167/1C	Fine wood charcoal from sample with ashes	910 ± 125	-24†	± 2‰	926 ± 129*
GX-4669	N10-2, LA 171/1C	Coarse wood charcoal	1175 ± 125	-24†	± 2‰	1191 ± 129*
GX-4670	N10-2, LA 177/1C	Wood charcoal	1045 ± 120	-24†	± 2‰	1061 ± 124*
GX-4671	N10-9, LA 207	Wood charcoal	1595 ± 130	-24†	± 2‰	1611 ± 134*
GX-4672	N10-9, LA 208	Wood charcoal	1495 ± 130	-24†	± 2‰	1511 ± 134*
GX-4673	N10-9, LA 209	Wood charcoal	1385 ± 185	-24†	± 2‰	1401 ± 188*

GX= Geochron Laboratories; Samples received 13 December 1976, reported 22 February 1977; OXA= Oxford Radiocarbon Accelerator Unit (ORAU), 2007

† From Stuiver and Polach 1977:Fig. 1; \*Estimated, based on Stuiver and Reimer 2015; blue calculated in this study

### Outlier Model Descriptions

As mentioned in the text, statistical techniques have been devised to accommodate and use outlier samples without having to remove them entirely from a model (allowing better integration and judgement of their fit—see S4 and S5). For this project, all samples were tagged as outliers, with the “good” samples fitted to the “General” model (at 0.05 probability) and samples identified as old wood fitted to the “Charcoal” model (at 1.0 probability-- i.e. 100%). Combined dates also used the “SSimple” model within the R\_Combine container (see Table S2 for exact usage).

































`Outlier_Model("General", T(5), U(0, 4), "t")`

This is the “General” model with priors drawn from Student’s t-distribution at 5 degrees of freedom [T(5)], in a uniform scale from 1-10,000 ( $10^4$ ), where the sample being measured could have a different date than the event (“t”) (Ramsey 2009a)

`Outlier_Model("Charcoal", Exp(1, -10, 0), U(0, 3), "t")`

This is the most robust model for charcoal samples— an exponential distribution with constants taken between -10 and 0, with the assumption that the date is at least >1 year older than the event but no more than 1000 ( $10^3$ ).

### S4. Outlier Analysis of N10-77/N10-12 Model (OxCal v.4.2.4)

Element	Ok 	Outlier 	Prior	Posterior	Model	Type
LA 2524 [OxA-18017(13)]			5	3	General	t
LA 2532 [OxA-18020(15)]			5	3	General	t
LA 2522 [OxA-18016(12)]			5	3	General	t
LA 2525 [OxA-18019(14)]			5	3	SSimple	s
LA 2525 [OxA-18018(14)]			100	100	SSimple	s
LA 1798 [OxA-17976(9)]			5	3	General	t
LA 1777 [OxA-17985(3)]			5	4	SSimple	s
LA 1777 [OxA-17970(3)]			5	4	SSimple	s
LA 1784 [OxA-17974(7)]			5	3	General	t
LA 1785/1 [OxA-17975(8)]			5	3	General	t
LA 1783 [OxA-17973(6)]			5	3	General	t
LA 1764 [OxA-17969(2)]			5	3	General	t
LA 1894/8 [OxA-18015(11)]			5	3	General	t
LA 1894/6 [OxA-18014(10)]			100	100	Charcoal	t
LA 1742 [OxA-17968(1)]			5	3	General	t

### S5. Outlier Analysis of N10-2 Model(OxCal v.4.2.4)

Element	Ok 	Outlier 	Prior	Posterior	Model	Type
LA 110/1C-Ork [GX-4662]			5	4	General	t
LA 115/2C-Tok [GX-4664]			5	4	General	t
LA 30/1C-Gom [GX-4659]			100	100	Charcoal	t
LA 177/1C-Gom [GX-4670]			5	4	General	t
LA 115/1C-Gom [GX-4663]			50	52	SSimple	s
LA 115/1C-Gom [OxA-18022 (18)]			100	100	SSimple	s
LA 34/1C-Gom [GX-4660]			100	100	SSimple	s
LA 34/1C-Gom [OxA-18021 (16)]			5	6	SSimple	s
LA 34/2C-Gom [GX-4661]			5	3	General	t
LA 171/1C-Bat [GX-4669]			100	100	Charcoal	t
LA 139/1C-Fern [GX-4666]			5	3	General	t

N10-77/N10-12 Final Model Agreement:


Amodel=89.3  
Aoverall=85.9

N10-2 Final Model Agreement:


Amodel=47.5  
Aoverall=54.7

### Previous Models

The final models for N10-2 and N10-77/N10-12 engaged the Charcoal and General outlier models in OxCal, along with a *Terminus Ante Quem* (TAQ) calendar date set for the entire sequence at AD 1544. Numerous variations of the model parameters were analyzed, but the effects of each change tended to be quite limited. A comparison of the original uniform model used on the N10-2 sequence with the final variation demonstrates the exiguous differences (Figures S6 and S7).




**S6. Original N10-2 Model, before Charcoal Outlier or AD 1544 TAQ applied (GX-4659 and GX-4666 removed)**




**S7. Final N10-2 Model, with Charcoal Outlier and AD 1544 TAQ**

## Trapezoidal Ceramic Models


Below are the full distributions produced using the trapezoidal models for the Tzunun (Figure S8), Terclerp (Figure S9) and Buk (Figure S10) ceramic phases. While it is tempting to include the trapezoid parameters in the main Bayesian models (e.g. Figure S7), they are, in fact, entirely separate processes. The chronological model is used to refine the sample dates (i.e. the event being dated), while the trapezoidal model combines associated sample dates into a gradual distribution (i.e. as a seriation). In order to use posteriors from the chronological model in the trapezoidal one, .prior files were inserted as references.



S8. Trapezoid Posteriors of Tzunun Ceramic Phase




**S9. Trapezoid Posteriors of Terclerp Ceramic Phase**



**S10. Trapezoid Posteriors of Buk Ceramic Phase**









# S11. Combined Trapezoid Posteriors of Ceramic Phases

## Calibration Curve Interactions



Interactions with the calibration curve were most noticeable in the Terminal Classic and Postclassic dates, with LA 1894/8 and LA 115/1C hitting long plateaus. Neither the precision of AMS dating nor Bayesian modelling could reduce the resultant spreads.




In the case of LA 34/1C, the combined date between GX-4660 and OxA-18021 created a bimodal distribution. Scrutiny of the provenience and stable isotope readings (see text) led to the “rejection” (i.e. assignment of a 1.0 outlier value) of GX-4660 due to old wood effects. Left combined with OxA-18021, however, GX-4660 still continued to pull OxA-18021 earlier enough to split the distribution between two points on the calibration curve. As mentioned in the text, the latest range (AD 1150-1220, CI: 44.3%) is probably the most correct, assuming the readings for OxA-18021 (charred maize) were closest to the timing of the caching event. However, the unmodeled date for OxA-18021 was AD 1150-1255 (CI 91.9%), meaning that the dates were actually *more accurate before their placement* within the modeled construction sequence. In this unique case, the Bayesian model may have had a negative effect on the accuracy of the final result!





In other cases, however, the combined precision of AMS measurements and Bayesian modelling worked perfectly, as can be seen in LA 2524 and LA 1784.



## Caching Locations




**S18.** Full Site Map (from Howie 2012)





(From Howie 2012: Plate VII.9)



S21. Eastern Profile of N-S Trench through N10-77/-12, showing N10-12/1, N10-12/2, the Boulders Phase, and N10-77

Lab # (Sample #)	Lot #	Structure	$\delta^{13}\text{C}$	Corrected Radiocarbon Age (BP), $1\sigma$	calAD (IntCal-13), $2\sigma$ (Unmodelled)	
					range	P
GX-4659	LA 30/1C	N10-2	-24 <sup>^</sup>	1786 ± 139*	90BC-AD565	95.40%
GX-4660	LA 34/1C	N10-2	-17.4*	915 ± 115	710-1225	95.40%
GX-4661	LA 34/2C	N10-2	-25.6*	830 ± 120	975-1395	95.40%
GX-4662	LA 110/1C	N10-2	-25.3*	1235 ± 130	550-1035	95.40%
GX-4663	LA 115/1C	N10-2	-27.7*	715 ± 130	1035-1435	95.40%
GX-4664	LA 115/2C	N10-2	-24 <sup>^</sup>	1251 ± 129*	545-1025	95.40%
GX-4665	LA 136/1C	N10-2	-26.2*	1690 ± 125	55-600	95.40%
GX-4666	LA 139/1C	N10-2	-24 <sup>^</sup>	826 ± 134*	905-1410	95.40%
GX-4667	LA 166	N10-7	-24 <sup>^</sup>	1526 ± 134*	215-770	95.40%
GX-4668	LA 167/1C	N10-2	-24 <sup>^</sup>	926 ± 129*	775-1385	95.40%
GX-4669	LA 171/1C	N10-2	-24 <sup>^</sup>	1191 ± 129*	605-1150	95.40%
GX-4670	LA 177/1C	N10-2	-24 <sup>^</sup>	1061 ± 124*	685-1215	95.40%
GX-4671	LA 207	N10-9	-24 <sup>^</sup>	1611 ± 134*	125-660	95.40%
GX-4672	LA 208	N10-9	-24 <sup>^</sup>	1511 ± 134*	215-775	95.40%
GX-4673	LA 209	N10-9	-24 <sup>^</sup>	1401 ± 188*	240-1020	95.40%
OxA-17968 (1)	LA 1742	N10-12	-25.9	1050 ± 24	900-925 960-1025	5.00% 90.40%
OxA-17969 (2)	LA 1764	N10-77	-26.8	1312 ± 25	655-725 740-770	70.40% 25.00%
OxA-17970 (3)	LA 1777	N10-77	-25.7	1409 ± 25	600-665	95.40%
OxA-17971 (4)	LA 1778	N10-77	-25.3	1423 ± 25	585-660	95.40%
OxA-17972 (5)	LA 1779	N10-77	-26.3	1367 ± 26	615-685	95.40%
OxA-17973 (6)	LA 1783	N10-77	-26.1	1280 ± 24	670-770	95.40%
OxA-17974 (7)	LA 1784	N10-77	-26.1	1304 ± 25	660-725 735-770	65.90% 29.50%
OxA-17975 (8)	LA 1785/1	N10-77	-26.7	1297 ± 25	660-730 735-770	63.10% 32.30%
OxA-17976 (9)	LA 1798	N10-77	-26.1	1284 ± 25	665-770	95.40%
OxA-17985 (3)	LA 1777	N10-77	-26.6	1402 ± 25	600-665	95.40%
OxA-18014 (10)	LA 1894/6	N10-12	-26.3	1282 ± 26	665-770	95.40%
OxA-18015 (11)	LA 1894/8	N10-12	-26	1206 ± 26	715-745 765-890	6.10% 89.30%
OxA-18016 (12)	LA 2522	N10-77	-26.2	1260 ± 26	665-780 790-805 810-825 840-865	90.50% 1.70% 0.90% 2.30%
OxA-18017 (13)	LA 2524	N10-77	-26.1	1275 ± 26	670-775	95.40%
OxA-18018 (14)	LA 2525	N10-77	-26.1	1331 ± 27	645-715 740-765	81.40% 14.00%
OxA-18019 (14)	LA 2525	N10-77	-26.1	1282 ± 26	665-770	95.40%
OxA-18020 (15)	LA 2532	N10-77	-28.3	1240 ± 26	685-780 785-875	64.50% 30.90%
OxA-18021 (16)	LA 34/1C	N10-2	-9.62	856 ± 25	1055-1255	95.40%
OxA-18022 (18)	LA 115/1C	N10-2	-26.2	950 ± 25	1020-1155	95.40%

Calibrated with OxCal v4.2.4 (Bronk Ramsey 2013)

IntCal13 northern atmospheric curve (Reimer et al. 2013); all calibrations rounded to 5

\*Estimated, based on Stuiver and Reimer 2015, see text and Table S3

<sup>^</sup>Based on Stuiver and Polach 1977